

Supplemental materials




Figure S1. Maturation functions for clearance (left), absorption rate constant (k_a , middle), and absorption mean transit time (MTT, right). Plotted are the percentage of mature values against post-menstrual age (for clearance) or post-natal age (for k_a and MTT). The blue line marks assumed gestational age at birth (9 months), while the red notches mark the ages of the study subjects.




Figure S2. Visual predictive check (VPC) of ethionamide concentration versus time after last dose (TALD), all observations on normal scale (left) and semi-logarithmic scale (right).




Figure S3. Visual predictive check (VPC) of ethionamide concentration versus time after last dose (TALD), stratified on whole (left) or crushed tablet (right).




Figure S4. Visual predictive check (VPC) of ethionamide concentration versus time after last dose (TALD), stratified on rifampicin co-administration.




Figure S5. Conditional weighted residuals (CWRES) versus time after last dose. The red line is a smooth function.

```

27 $PROBLEM      Ethionamide paediatric PK
28 $INPUT        ID CDATE=DROP CTIME=DROP PTIME=DROP AMT ADM FORM DV BLQ
29              ORIGOCC EVENT=DROP WT HT SEX ETN HIV AGE FFM EVID OCC TIME
30              TALD STUDY HAZ WAZ BAZ INH RIF PZA EMB AMI OFL TRD PAS D4T
31              TTC EFV RTV KAL PYR AZT
32 $DATA         ../Data/nmset19_noSS.csv IGNORE=@
33 $SUBROUTINE    ADVAN13 TRANS1 TOL=9 ATOL=6
34 $MODEL        NCOMPARTMENTS=2 COMP=(ABS DEFDOSE)
35              COMP=(CENTRAL DEFOBSERVATION)
36 $ABBREVIATED  REPLACE ETA(OCC_BIO)=ETA(1,2,3,4,5,6)
37 $ABBREVIATED  REPLACE ETA(OCC_MTT)=ETA(7,8,9,10,11,12)
38 $ABBREVIATED  REPLACE ETA(OCC_KA)=ETA(13,14,15,16,17,18)
39 $PK
40
41 ;;; ABSAGE-DEFINITION START
42 ABSAGE = ( 1 + THETA(14)*EXP(-AGE))
43 ;;; ABSAGE-DEFINITION END
44
45 ;;; BIOHIV-DEFINITION START
46 IF(HIV.EQ.0) BIOHIV = 1 ; Most common
47 IF(HIV.EQ.1) BIOHIV = ( 1 + THETA(11))
48 ;;; BIOHIV-DEFINITION END
49
50 ;;; MTT
51 IF(ADM.EQ.2) TVMTT = THETA(9) ; Oral
52 IF(ADM.EQ.1) TVMTT = THETA(9)*(1 + THETA(10)) ; Nasogastric tube
53 ;;; MTT
54
55 ;;; CLRIF-DEFINITION START
56 IF(RIF.EQ.0) TVCL = THETA(7) ; Most common
57 IF(RIF.EQ.1) TVCL = THETA(7)*(1 + THETA(13))
58 ;;; CLRIF-DEFINITION END
59
60 ;;; KA-RELATION START
61 KACOV=ABSAGE
62 ;;; KA-RELATION END
63
64 ;;; MTT-RELATION START
65 MTTCOV=1 / ABSAGE
66 ;;; MTT-RELATION END
67
68 ;;; BIO-RELATION START
69 BIOCOV=BIOHIV
70 ;;; BIO-RELATION END
71
72
73 ; ----- Typical values of covariates
74 TVWT = 13
75 TVFAT = 2.7
76 TVFFM = 10.5
77 TVPERFAT = 0.27
78
79 ; Maturation of CL
80 PMAGE = AGE + 9/12
81
82 LOGPMAGE50 = THETA(1)
83 GAMMA = EXP(THETA(2))

```

```

84
85 MATCL=1/(1+EXP(-GAMMA*(LOG(PMAGE)-LOGPMAGE50)))
86
87 IOV_BIO      =      ETA(OCC_BIO)
88 IOV_MTT      =      ETA(OCC_MTT)
89 IOV_KA       =      ETA(OCC_KA)
90
91 IIV_V        =      ETA(19)
92 IIV_CL       =      ETA(20)
93
94 ALLMCL_WT    =      (WT/TVWT)**0.75
95 ALLMV_WT     =      (WT/TVWT)
96
97 ADDERR       =      THETA(3)
98 PROPERR      =      THETA(4)
99 TVBIO        =      THETA(5)
100 TVBIO        =      TVBIO * BIOCOV
101 TVV          =      THETA(6) * ALLMV_WT
102 TVCL         =      TVCL * ALLMCL_WT * MATCL
103 TVNN         =      THETA(8)
104 TVNN         =      TVNN
105 TVMTT        =      TVMTT * MTTCOV
106 TVKA         =      THETA(12)
107 TVKA         =      TVKA * KACOV
108
109
110 BIO          =      TVBIO*EXP(IOV_BIO)
111 V            =      TVV*EXP(IIV_V)
112 CL           =      TVCL*EXP(IIV_CL)
113 NN           =      TVNN
114 MTT          =      TVMTT*EXP(IOV_MTT)
115 KTR          =      (NN+1)/MTT
116 KA          =      TVKA*EXP(IOV_KA)
117 K            =      CL/V
118 S2          =      V
119
120 ; Transit compartment code
121 F1           =      0
122
123 IF (NEWIND/=2.OR.EVID>=3) THEN ; new individual, or reset event
124     ; The values read here will be stored in TDOS and PD in this very PK call.
125     TNXD=TIME ; Time of the dose
126     PNXD=AMT ; Amount. If it's zero, the DE is deactivated.
127 ENDIF
128
129 TDOS=TNXD ; This will either save here the temporary values
130     ; if it's a new individual...
131 PD=PNXD ; ...or the values which were read one record ahead
132     ; during the execution of the previous record.
133
134 IF(AMT>0) THEN ; This reads one record ahead and stores
135     ; the data to be used when running the following record
136     TNXD=TIME
137     PNXD=AMT
138 ENDIF
139
140 ; To speed up the computation, I calculate here all the

```

```

141 ; non-time-varying quantities used in $DES
142 PIZZA = LOG(BIO*PD*KTR + 0.00001) - GAMLN(NN+1)
143
144
145 $DES
146
147 TEMPO = T-TDOS ; this is time after dose for the transit, it should
148                 ; always be >= 0
149
150 KTT = 0
151 DADT(1) = -KA*A(1)
152
153 IF(PD.GT.0.AND.TEMPO.GT.0) THEN ; This happens only id PD>0, so
154                                 ; only if a dose has been detected
155     KTT = KTR*(TEMPO)
156     DADT(1) = EXP(PIZZA+NN*LOG(KTT)-KTT) -KA*A(1)
157 ENDIF
158
159 DADT(2) = KA*A(1) -K*A(2)
160
161
162 $ERROR
163
164 IPRED          =      A(2)/V
165 IRES           =      DV-IPRED
166 PROP          =      IPRED*PROPERR
167 ADD           =      ADDERR
168 W             =      SQRT(ADD**2+PROP**2)
169
170 IF (W.LE.0.000001) W=0.000001
171
172 IWRES          =      IRES/W
173 Y              =      IPRED + W*ERR(1)
174
175 IF (ICALL.EQ.4.AND.Y.LE.0.01) Y=0.01
176
177 AA1 = A(1)
178 AA2 = A(2)
179
180 $THETA
181 -0.132497 ; LOGPMAGE50
182 0.954688 ; GAMMA
183 (0.00626,0.0138181) ; AddErr
184 (0,0.171622) ; PropErr
185 1 FIX ; BIO
186 (0,21.4485) ; V/F
187 (0,8.88233) ; CL/F
188 (0,1.309) ; NN
189 (0,1.06536) ; MTT_O
190 (-1,-0.678872) ; MTT_NGT
191 (-1,-0.222796,5) ; BIOHIV1
192 (0,0.867349) ; KA
193 (0,0.159957) ; CLRIF1
194 (0,1.43685,2) ; ABSAGE
195 $OMEGA BLOCK(1)
196 0.0621268 ; IOV_BIO
197 $OMEGA BLOCK(1) SAME

```

```

198 $OMEGA BLOCK(1) SAME
199 $OMEGA BLOCK(1) SAME
200 $OMEGA BLOCK(1) SAME
201 $OMEGA BLOCK(1) SAME
202 $OMEGA BLOCK(1)
203 0.940804 ; IOV_MTT
204 $OMEGA BLOCK(1) SAME
205 $OMEGA BLOCK(1) SAME
206 $OMEGA BLOCK(1) SAME
207 $OMEGA BLOCK(1) SAME
208 $OMEGA BLOCK(1) SAME
209 $OMEGA BLOCK(1)
210 0.131732 ; IOV_KA
211 $OMEGA BLOCK(1) SAME
212 $OMEGA BLOCK(1) SAME
213 $OMEGA BLOCK(1) SAME
214 $OMEGA BLOCK(1) SAME
215 $OMEGA BLOCK(1) SAME
216 $OMEGA 0 FIX ; IIV_V
217 0.0289429 ; IIV_CL
218 $SIGMA 1 FIX
219
220 $ESTIMATION MSFO=final.msf MAXEVAL=9999 PRINT=1 METHOD=1 INTER
221 NOABORT NSIG=3 NONINFETA=1 ETASTYPE=1 SADDLE_RESET=1
222 SIGL=9
223 $COVARIANCE PRINT=E
224 ;$SIMULATION (12345) ONLYSIMULATION
225
226
227 NONMEM Model Code. NONMEM model code for the final model.
228
229

```